
A METAFONT for rustic capitals

Victor Sannier

1 Introduction

Littera capitalis rustica (literally country capital let-
ters) are a script of the Latin alphabet, the earliest
examples of which date back to the first century CE,
notably on election posters in the city of Pompeii
(see Figure 1). They were gradually regularised in
the fourth and fifth centuries and remained in use
until the ninth and tenth centuries, but mostly for
titles and distinctions [1].

Figure 1: Election poster in Asellina’s tavern, Pompeii

Many manuscripts contain splendid examples
of rustic capitals. These include the Vergilius Vat-
icanus [7] and the fifth-century Vergilius Romanus
and Codex Mediceus. The first one, an incomplete
copy of Virgil’s Aeneid and Georgics, will be the main
reference for our work, although we will attempt to
rationalise its handwriting with the METAFONT sys-
tem [3] — while maintaining a sense of authenticity —
rather than to reproduce it exactly, imperfection for
imperfection. Thus, in the spectrum of type revivals
described by Olocco and Patanè [6, Chapter 1], which
ranges from literal reproduction to reinvention, we
intend to occupy a middle ground.

Whenever possible, we use the same terminology
as that found in [2, Chapter 2].

2 Design of the repeating components

Our METAFONT project is divided into several files,
ruscap.mf defines the macros and characters, while

ruscap10.mf, ruscap14.mf, &c. define the dimen-
sions to be used for rendering to a particular font
size. In the rest of this article all values are taken
from ruscap10.mf.

2.1 Lengths and angles

After numerous measurements on various fragments
of the manuscript that we use as a reference, it
appeared that the vertical space could be divided
into twelve units, from which we also define s (for
sidebearing) and o (for overshoot).

u# := 10/12 pt#; % unit
s# := 3/4 u#; % sidebearing
o# := 1/4 u#; % overshoot

Then, most capitals occupy nine units above the
baseline, some eleven (such as ‘F’ and ‘L’), and the
letter ‘Q’ (and ‘G’ in one variant) goes down one
unit below it. The crossbar in ‘E’, the junction of
the two lobes of ‘B’, &c. are slightly above half of
the upper space.

cap_height# := 9 u#;
asc_height# := 11 u#;
desc_height# := 1 u#;
crossbar_height# := 5 u#;

Next, the serifs in our font have the shape of
a tilde, so all we need to specify is a width and an
angle (0 would mean the serif is just a horizontal
stroke); see the next section and [3, p. 152].

serif_width# := 5/2 u#;
serif_angle := 90 / 6;

Two dimensions are also needed for components
of some characters, namely the angle of the stroke
in the letters ‘A’, ‘N’, &c., and the width and angle
of the spurs.

diag_angle := 90 + 35;
spur_width# := 1/2 u#;
spur_angle := 0;

Finally, the virtual pencil used is defined by
the thickness of the thickest and thinnest strokes it
can draw, and by its slope, here 35 degrees. Rustic
capitals are known to be written with a very inclined
tool [1].

thick# := 5/4 u#;
thin# := 1/3 u#;
pen_angle := 65; % 90 - 35

2.2 Macros

Let us give the code of some macros we have written
to improve consistency and limit repetition in the
project.

2.2.1 draw_serif

To create a serif, we constrain its ends to be hori-
zontally aligned and serif_width apart, and draw
a line between them with the same initial and final
angles.

def draw_serif(suffix i, j)(expr width) =
rt x.j - lft x.i = width;
y.i = y.j;
draw z.i{dir serif_angle}

.. {dir serif_angle}z.j;
enddef;

See, for example, the top arm of the letter ‘F’
in Figure 2.

2.2.2 draw_diag_stroke

With the spurs defined at both ends of the stroke,
all we need to do is set the angle between them and
connect all the points. We have increased the tension
in the main part so that it is almost straight but still
smoothly connects the spurs.

def draw_diag_stroke(suffix i, j)(expr a) =
z.i - z.i.l = z.j.r - z.j

= spur_width * dir 0;
z.j - z.i = whatever * (dir angle);
draw z.i.l .. z.i

.. tension 3

.. z.j .. z.j.r;
enddef;

In most cases the angle parameter will take
the value diag_angle, but this is not always the
case. For example, the two diagonal strokes in the
letter ‘M’ don’t have the same angle.

2.2.3 draw_I

We use the same code to draw the letters ‘I’ and ‘L’,
and the left part of the letters ‘B’, ‘P’, ‘R’, &c.

def draw_I(suffix i, j, k, l)(expr sw) =
x.i = x.j; % vertical stem
top y.i = h; bot y.j = 0;
z.i - z.i.l

= spur_width * dir spur_angle;

% Serif
rt (2 x.j - x.k) - lft x.k

= serif_width;
y.j = y.k;
draw_serif(k, l)(sw);

draw z.i.l .. z.i .. tension 5 .. z.j;
enddef;

The way we constrain the position of the serif
ends may require some explanation. The sw vari-
able stores the total width of the serif to be drawn.
Generally it is not centred around the stem; the
right part can be as long as needed and only the
width of the left part is constant and should be half
the value s of serif_width. To achieve this, the
following equation should hold:

xk + 2(xj − xk) = xk + s

which, after rewriting and taking into account the
size of the pencil nib, gives the above code.

3 Design of the characters and kerning

3.1 Example of the letter ‘T’

beginchar("T", 6u# + 2s#, cap_height#, 0);
"Rustic␣T";
pickup rustic_pen;
x1 = w - x2; top y1 = h;
draw_serif(1, 2)(w - 2s);
x3 = w - x4; bot y3 = 0;
draw_serif(3, 4)(serif_width);
draw 1/2 [z1, z2] .. 1/2 [z3, z4];
labels(range 1 thru 4);

endchar;

3.2 Example of the letter ‘N’

The letters ‘M’ and ‘N’ are the most involved we
have designed, but the code is still straightforward.

beginchar("N", 7u# + 2s#, cap_height#, 0);
"Rustic␣N";
pickup rustic_pen;

% Diagonal stroke
x1 + x2 = w; top y1 = h; bot y2 = 0;
draw_diag_stroke(1, 2)(diag_angle);

% Left stem
bot y3 = 0; lft x3 = s;
draw z1 .. z3;
1/2 (z3.l + z3.r) = z3;
draw_serif(3.l, 3.r, serif_width);

% Right stem
x4 = x5 = x2.r; % vertical stem
top y4 = h; bot y5 = 0;
z4 - z4.l

= spur_width * dir spur_angle;
draw z4.l .. z4 -- z5;

labels(range 1 thru 5);
endchar;

Perhaps the most debatable choice we have made
is the angled left-hand stem and straight right-hand
stem, but we think it works well with letters such as
‘A’ and ‘L’.

3.3 Example of the letter ‘S’

The design of the letter ‘S’ is peculiar in that it
consists of a single stroke and doesn’t call any of our
custom macros.

beginchar("S", 4u# + s#, cap_height#, 0);
"Rustic␣S";
pickup rustic_pen;
lft x2 = s; x2 = x4;
rt x1 = w; x1 = x3;
top y1 = h - u; bot y4 = o;
h - y2 = y3;
z2 - z3 = whatever * dir diag_angle;
draw z1{curl 2} .. z2

.. z3 .. {curl 1}z4;
endchar;

Note the slight asymmetry introduced by a dif-
ferent curl value at each end.

3.4 Side-bearing and kerning

We chose a uniform side-bearing throughout the type-
face and corrected glaring kerning problems with a
ligature table.
ligtable "A": "C" kern -.5u#, "T" kern -u#;
ligtable "K": "O" kern -.5u#;
ligtable "L": "O" kern -.5u#, "T" kern -u#;
ligtable "N": "V" kern .5u#;
...

We believe that fine-tuning the interletter spac-
ing does not align with our goal of authenticity.

3.5 Remarks

While scanning various manuscripts and online re-
sources, we came across two variants of the letter ‘G’
[5, 8]. In the final design, we chose the one that looks
more like a square capital as the main variant, and
made the other available as ‘g’.

4 Comparison with other typefaces

See Figures 3, 4 and 5 for a comparison of fragments
of Vergilius Vaticanus with the typefaces designed
by (a) Landers [4], (b) Wilson [9] and (c) the author
respectively. The (b) and (c) typefaces were created
with the METAFONT system, (a) was not.

While the former designs may be well suited
to their authors’ aims, they do not correspond to

ABCDEFG
HIKLM
NOPQRS
TVXYZ

Figure 2: Character set of the ruscap typeface

locusinqui
locusinqui
LOCVSINQVI

Figure 3: The words “locus in qui” rendered in
different typefaces

the one we stated in the introduction. In particular,
we would like to draw the reader’s attention to the
following points:

• in (a), the letters are slightly angled to the right,

• in (a), the letters ‘E’ and ‘N’ are not the same
height, and neither are the letters ‘S’ and ‘U’,

• in (a), the letters ‘D’ and ‘R’ feature a large
spur,

• in (b), the crossbar of the letter ‘E’ is long and
sinuous,

• in (b), the strokes become significantly thicker
as they descend,

• in (a) and (b), some paths are quite steep, such
as the letter ‘D’ in (a) and the letter ‘G’ in (b).

lacrimasque
lacrimasque
LACRIMASQVE

Figure 4: The word “lacrimasque” rendered in
different typefaces

undegenusducis
undegenusducis
VNDEGENVSDVCIS

Figure 5: The words “unde genus ducis” rendered in
different typefaces

It also seems to us that our METAFONT code is
simpler than that written by Wilson, partly, but not
only, because our character set is smaller.

5 Conclusion and future work

In this article, we have limited ourselves to presenting
just one font from the ruscap family, but META-
FONT makes it easy to achieve a variety of designs,
including different weights, by changing just a few
parameters. For example, the result of setting a
uniform pen thickness, a serif angle of 0 and a slightly
higher crossbar height is shown in Figure 6.

In the near future, we plan to iterate on the
shapes of some characters (particularly ‘M’ and ‘U’),
consider the revisions made during the TUG 2023
event, and finally submit our typeface to the Compre-
hensive TEX Archive Network (CTAN) for everyone
to use freely.

We will also consider extending the character
set to include the two “Ramist letters” ‘J’ and ‘U’,
the letter ‘W’ and the Indo-Arabic digits.

CICERONIANVS
CICERONIANVS

Figure 6: The word ‘ciceronianus’ rendered in two
fonts of the ruscap family

